

Implementing Wake-Up on Key Stroke

INTRODUCTION

In certain applications, the PIC16CXX is exercised only when a key is pressed, eg. remote keyless entry. In such applications, the battery life can be extended by putting the PIC16CXX to sleep during the inactive state and when a key is pressed, the PIC16CXX wakes up, does the task, and then goes back to sleep.

IMPLEMENTATION

The circuit in Figure 1 depicts an application with two keys. The PIC16C54 is normally in SLEEP mode consuming very little operating current. If either of the two keys is pressed, the PIC16C5X 'wakes up', scans the keys and turns on one of the two LED's. When SW1 is pressed, the green LED is turned on and when SW2 is pressed the red LED is turned on. The LED's are used purely for demonstration purposes. In real life application, a transmission will be completed before putting the PIC16C5X back in sleep. This example can be extended to handle more than two keys.

In the sleep mode, the scan outputs (SCAN1 and SCAN2) are both set to a low logic level. In this state, the capacitor C is fully charged and a high logic level is present at the MCLR pin of the PIC16C5X. When a key is pressed, C discharges through either R2 or R3 (de-


pending on SW1 or SW2 being pressed) and the voltage across C falls rapidly (approx. 1 ms), causing a low at the MCLR pin of the PIC16C5X, which in turn causes the PIC16C5X to wake up and enter its reset state. In reset, the SCAN1 and SCAN2 outputs default to a high impedance mode, so the discharge path for capacitor C is blocked and it charges to a high level through resistor R1. Note that the RC values have been chosen such, that the discharge and charge cycles times are less than the reset time for the PIC16C5X (approx. 18 ms), and certainly far less than the minimum duration of a key-press (approx. 50-100 ms).

After the reset cycle is completed, the code execution momentarily takes the SCAN1 and SCAN2 outputs low in order to sample the key stroke(s). This does not cause the capacitor to discharge since the duration of the low is of the order of 10 micro seconds.

Once the keystroke function has been executed, the program loops until the key has been released, sets the SCAN1 and SCAN2 outputs low and "goes back to sleep". Resistors R4-R8 are not required for functionality, but are recommended to provide protection from electrostatic discharge (ESD). Switches SW1 and SW2, when pressed may frequently pass ESD to the PIC16C54.

*Author: Stan D'Souza
Logic Products Division*

FIGURE 1 - TWO KEY INTERFACE TO PIC16C5X



Implementing Wake-Up on Key Stroke

FIGURE 2 - TWO KEY SCAN/WAKE-UP TIMING DIAGRAM





FIGURE 3 - PIC16C5X INTERFACE TO 4 X 4 KEY MATRIX



Implementing Wake-Up on Key Stroke

MPASM 1.00 Released
Key Stroke Wake Up

WU.ASM 7-15-1994 13:10:54

PAGE 1

```
LOC OBJECT CODE LINE SOURCE TEXT
0001 TITLE "Key Stroke Wake Up"
0002 LIST P = 16C54,f=inhx8m
0003 ;*****
0004 ; Program demonstrating key stroke wake up for
0005 ; the PIC16CXX. Program has been implemented for
0006 ; two keys, but can be extended for more keys.
0007 ; When SW1 is pressed a green LED lights up.
0008 ; When SW2 is pressed a red LED lights up.
0009 ;*****
0010 ;
0011 ; Define equates
0012 ;
0013 PC EQU 2
0014 PORT_B EQU 6
0015 SCAN1 EQU 2
0016 SCAN2 EQU 3
0017 SW1 EQU 0
0018 SW2 EQU 1
0019 GRN_LED EQU 4
0020 RED_LED EQU 5
0021 MSEC_20 EQU D'20'
0022 DB1 EQU 8
0023 GP EQU 8
0024 DB2 EQU 9
0025 ;
0026 ;PORT_B ASSIGNMENTS:
0027 ; 0 -> SW1 INPUT
0028 ; 1 -> SW2 INPUT
0029 ; 2 -> SCAN1 OUTPUT
0030 ; 3 -> SCAN2 OUTPUT
0031 ; 4 -> GRN_LED OUTPUT
0032 ; 5 -> RED_LED OUTPUT
0033 ; 6&7 -> ASSIGNED AS DUMMY OUTPUTS
0034 ;
0035 ;
0036 ;
0037 ORG 0
0038 ;
0039 START
0040 CALL INIT_PORT_B ;INITIALIZE PORT B
0041 CALL DELAY ;DELAY 20 MSECS
0042 CALL SCAN_KEYS ;GET KEY VALUES
0043 MOVWF GP ;SAVE IN RAM
0044 BTFSC GP,SW1 ;SKIP IF SW1 NOT PRESSED
0045 CALL TURN_GREEN_ON ;ELSE DO ROUTINE
0046 BTFSC GP,SW2 ;SKIP IF SW2 NOT PRESSED
0047 CALL TURN_RED_ON ;ELSE DO ROUTINE
0048 CHK_FOR_KEY
0049 CALL DELAY ;DELAY FOR 20 MSEC
0050 CALL SCAN_KEYS ;GET KEY HIT
0051 XORLW 0 ;EXCL. OR WITH 0
0052 BNZ CHK_FOR_KEY ;KEY STILL PRESSED
0053 ;THEN LOOP
0054 NO_KEY_PRESSED
0055 BCF PORT_B,SCAN1 ;SET SCAN LINES LOW
0056 BCF PORT_B,SCAN2 ; /
0057 SLEEP ;SLEEP
0058 ;
0059 ;
0060 ;
0061 INIT_PORT_B
0062 MOVLW B'00000011' ; config RB0, 1 as i/p's
0063 TRIS PORT_B ; and RB2-7 as o/p's
0064 MOVLW 0FFh
0065 MOVWF PORT_B ;DEFAULT VALUES FOR PORT_B
0066 RETLW 0 ;RETURN WITH NO ERROR
```

2

Implementing Wake-Up on Key Stroke

```
0067 ;
0068 ;This routine, scans two keys and returns the following:
0069 ;    0 if no key is pressed
0070 ;    1 if SW1 is pressed
0071 ;    2 if SW2 is pressed
0072 ;    3 if SW1 and SW2 are pressed
0073 ;
0074 SCAN_KEYS
0015 0446    0075      BCF      PORT_B,SCAN1    ;ENABLE SCAN FOR SW1
0016 0466    0076      BCF      PORT_B,SCAN2    ;EABLE SCAN FOR SW2
0017 0C03    0077      MOVLW   B'00000011'      ;LOAD MASK IN W
0018 0146    0078      ANDWF   PORT_B,0        ;AND WITH PORT
0019 0546    0079      BSF      PORT_B,SCAN1    ;DISABLE SCAN
001A 0566    0080      BSF      PORT_B,SCAN2    ;
001B 01E2    0081      ADDWF   PC,1            ;GET OFFSET TO TABLE
001C 0803    0082      RETLW   3                ;SW1 AND SW2 PRESSED
001D 0802    0083      RETLW   2                ;SW2 PRESSED
001E 0801    0084      RETLW   1                ;SW1 PRESSED
001F 0800    0085      RETLW   0                ;NO KEY PRESSED
0086 ;
0087 ;DELAY, IS A APPROX. WAIT FOR 20.4mSECS, FOR A SYSTEM
0088 ;USING A 2 Mhz CRYSTAL CLOCK.
0089 DELAY
0020 0C14    0090      MOVLW   MSEC_20
0021 0028    0091      MOVWF   DB1
0022 0069    0092      DLY1
0023 02E8    0093      CLRF    DB2
0024 0A26    0094      DECFSZ  DB1
0025 0800    0095      GOTO    DLY2
0026 02E9    0096      RETLW   0
0027 0A26    0097      DLY2
0028 0A22    0098      DECFSZ  DB2                ;INNER LOOP = 1.02 MSEC.
0099      GOTO    DLY2                ;
0100      GOTO    DLY1
0101 ;
0102 ;
0103 TURN_GREEN_ON
0029 0486    0104      BCF      PORT_B,GRN_LED
002A 0800    0105      RETLW   0
0106 ;
0107 TURN_RED_ON
002B 04A6    0108      BCF      PORT_B,RED_LED
002C 0800    0109      RETLW   0
0110 ;
0111      END
0112
0113
```

MEMORY USAGE MAP ('X' = Used, '-' = Unused)

```
0000 : XXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXX- _____
0040 : _____
```

All other memory blocks unused.

```
Errors   :    0
Warnings :    0
```

WORLDWIDE SALES & SERVICE

AMERICAS

Corporate Office

Microchip Technology Inc.
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 602 786-7200 Fax: 602 786-7277
Technical Support: 602 786-7627
Web: <http://www.mchip.com/microhip>

Atlanta

Microchip Technology Inc.
500 Sugar Mill Road, Suite 200B
Atlanta, GA 30350
Tel: 770 640-0034 Fax: 770 640-0307

Boston

Microchip Technology Inc.
5 Mount Royal Avenue
Marlborough, MA 01752
Tel: 508 480-9990 Fax: 508 480-8575

Chicago

Microchip Technology Inc.
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 708 285-0071 Fax: 708 285-0075

Dallas

Microchip Technology Inc.
14651 Dallas Parkway, Suite 816
Dallas, TX 75240-8809
Tel: 214 991-7177 Fax: 214 991-8588

Dayton

Microchip Technology Inc.
35 Rockridge Road
Englewood, OH 45322
Tel: 513 832-2543 Fax: 513 832-2841

Los Angeles

Microchip Technology Inc.
18201 Von Karman, Suite 455
Irvine, CA 92715
Tel: 714 263-1888 Fax: 714 263-1338

New York

Microchip Technology Inc.
150 Motor Parkway, Suite 416
Hauppauge, NY 11788
Tel: 516 273-5305 Fax: 516 273-5335

AMERICAS (continued)

San Jose

Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408 436-7950 Fax: 408 436-7955

ASIA/PACIFIC

Hong Kong

Microchip Technology
Unit No. 3002-3004, Tower 1
Metroplaza
223 Hing Fong Road
Kwai Fong, N.T. Hong Kong
Tel: 852 2 401 1200 Fax: 852 2 401 3431

Korea

Microchip Technology
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku,
Seoul, Korea
Tel: 82 2 554 7200 Fax: 82 2 558 5934

Singapore

Microchip Technology
200 Middle Road
#10-03 Prime Centre
Singapore 188980
Tel: 65 334 8870 Fax: 65 334 8850

Taiwan

Microchip Technology
10F-1C 207
Tung Hua North Road
Taipei, Taiwan, ROC
Tel: 886 2 717 7175 Fax: 886 2 545 0139

EUROPE

United Kingdom

Arizona Microchip Technology Ltd.
Unit 6, The Courtyard
Meadow Bank, Furlong Road
Bourne End, Buckinghamshire SL8 5AJ
Tel: 44 0 1628 851077 Fax: 44 0 1628 850259

France

Arizona Microchip Technology SARL
2 Rue du Buisson aux Fraises
91300 Massy - France
Tel: 33 1 69 53 63 20 Fax: 33 1 69 30 90 79

Germany

Arizona Microchip Technology GmbH
Gustav-Heinemann-Ring 125
D-81739 Muenchen, Germany
Tel: 49 89 627 144 0 Fax: 49 89 627 144 44

Italy

Arizona Microchip Technology SRL
Centro Direzionale Colleoni
Palazzo Pegaso Ingresso No. 2
Via Paracelso 23, 20041
Agrate Brianza (MI) Italy
Tel: 39 039 689 9939 Fax: 39 039 689 9883

JAPAN

Microchip Technology Intl. Inc.
Benex S-1 6F
3-18-20, Shin Yokohama
Kohoku-Ku, Yokohama
Kanagawa 222 Japan
Tel: 81 45 471 6166 Fax: 81 45 471 6122

9/22/95

All rights reserved. © 1995, Microchip Technology Incorporated, USA.

Information contained in this publication regarding device applications and the like is intended through suggestion only and may be superseded by updates. No representation or warranty is given and no liability is assumed by Microchip Technology Incorporated with respect to the accuracy or use of such information, or infringement of patents or other intellectual property rights arising from such use or otherwise. Use of Microchip's products as critical components in life support systems is not authorized except with express written approval by Microchip. No licenses are conveyed, implicitly or otherwise, under any intellectual property rights. The Microchip logo and name are registered trademarks of Microchip Technology Inc. All rights reserved. All other trademarks mentioned herein are the property of their respective companies.